

18. Fraction Towers

General Description

This program is intended mainly for use with younger children. The instructions will be displayed on the screen at the start of the program. Press any key to start the game. The screen will contain a series of fraction towers, all of which will be blank except the top one. The object is to place increasingly large fractions down the fraction tower.

Detailed Description

Lines 60-80 Set envelope for sound.
90 Check for escape key
110-140 Display title and instructions.
150-170 GOTO procedure to display boxes on screen.
180-200 Read next 'starting fraction' in, and check for terminator.
210-260 Display fraction and get input from keyboard.
270-280 Is it the last fraction on the tower? Go back and display next fraction if not.
290-350 Does user want another game?
360-430 Display boxes on the screen.
440-470 Display fraction on the screen.
480-550 Prompt for next fraction from keyboard.
560-630 Check for valid input.
640-700 Display invalid input and cue sound.
710-740 Set D & N = to denominator and numerator.
750 Is denominator = 0 ($1/0$ - infinity), if so show invalid input.
760 Is denominator too large (i.e. >999)?
770-780 Are N or D integers? If not, display message.

790-810 Are fractions too small? The same? Or 1?

820-870 Accept fraction.

880-930 Display error - i.e. too small, the same, the same as

1.

940-980 Display title.

990-1010 Cue sound.

1020-1030 Wait for key to be pressed.

1040-1210 Print instructions at correct position and set A\$ =
to the instructions to be printed. GOTO procedure and display
A\$.

Educational Note

This program should really only be used once you have taught fraction towers in the classroom. It provides an entertaining medium for experimenting with fraction towers. I suspect the program is better used with primary children rather than secondary children, though I have used it with first years in secondary school.

I found that a group of four or five round a keyboard and monitor happily discussed amongst themselves what the next fraction tower ought to be.

Program Listing

```
>
10 REM *****
20 REM *          FRACTION TOWERS          *
30 REM *      ADAPTED FOR THE BBC-B      *
40 REM * BY ANDREW PUSEY FEB 1983 *
50 REM *****
60 REM SET UP ENVELOPES FOR SOUND
70 ENVELOPE 1,1,100,-100,0,20,20,40,124,-4,-1,-5,126,126
80 ENVELOPE 2,1,4,-4,-1,25,25,40,123,0,-1,-1,126,126
90 REM HAS THE USER PRESSED 'ESCAPE'
100 ON ERROR GOSUB 1230
110 REM DISPLAY MAIN TITLE AND
120 REM INSTRUCTIONS
130 MODE7
140 PROC_TITLE
150 REM SET UP BOXES ON SCREEN
160 CLS
170 PROC_SCREEN
180 REM READ STARTING FRACTION
190 READ F$,ANS
200 IF F$="END"THEN PROC_FINISH
210 REM SET HEIGHT OF TOWER TO ZERO
220 H=0
```

```

230 REM GO AND DISPLAY FRACTION
240 PROC_DISPLAY
250 REM GO AND ASK FOR USER'S FRACTION
260 PROC_KEYBOARD
270 REM IS THE TOWER COMPLETE
280 IF H<>4THEN240
290 REM USER HAS COMPLETED TOWER
300 CLS
310 PRINT "'CHR$(131);CHR$(141);"WELL DONE. TRY ANOTHER
?"
320 PRINT CHR$(131);CHR$(141);"WELL DONE. TRY ANOTHER ?"
330 YN$=GET$
340 IF LEFT$(YN$,1)<>"Y"THEN PROC_FINISH
350 GOTO 160
360 DEFPROC_SCREEN
370 REM DISPLAY BOXES ON SCREEN
380 FOR A=0TO4
390 PRINT TAB(4,A*4+2);CHR$(130);"*****"
400 PRINT TAB(4,A*4+3);CHR$(130);"*"
410 PRINT TAB(4,A*4+4);CHR$(130);"*****"
420 NEXTA
430 ENDPROC
440 DEFPROC_DISPLAY
450 REM DISPLAY FRACTION ON SCREEN
460 PRINT TAB(6,H*4+3);CHR$(131);F$;CHR$(130)
470 ENDPROC
480 DEFPROC_KEYBOARD
490 REM GET FRACTION FROM USER
500 PRINT TAB(8,(H+1)*4+3);CHR$(134);CHR$(136);"?";CHR$(13
7);CHR$(130)
510 PRINT TAB(4,22);SPC(33)
520 PRINT TAB(4,22);CHR$(131);
530 INPUT"Next Fraction (X/Y)",NF$
540 PRINT TAB(16,16);" "
550 PRINT TAB(16,17);" "
560 T=0:QQ=0:REPEAT
570 T=T+1
580 IF LEN(NF$)<3THENQQ=T:T=4:GOTO610
590 X$=MID$(NF$,T,1)
600 IFX$="/"THENQQ=T:T=5
610 UNTIL T>3
620 IF T=5 THEN T=QQ:GOTO730
630 IF QQ<>0THEN T=QQ
640 REM WRONG USER INPUT
650 PRINT TAB(5,23);CHR$(129);"INVALID INPUT"
660 SOUND1,2,100,10
670 A$=INKEY$(200)
680 T=0
690 PRINT TAB(5,23);SPC(30)" "
700 GOTO 510
710 REM CHANGE STRING TO NUMERATOR
720 REM AND DENOMINATOR VALUES
730 N=VAL(LEFT$(NF$,T))
740 D=VAL(RIGHT$(NF$,LEN(NF$)-T))
750 IF D=0THEN 650
760 IF D>999 THEN 650
770 REM VALIDATE USER FRACTION
780 IFN<>INT(N)ORD<>INT(D)THENA$="NO POINT":PROC_ERROR:GOT
O510
790 IF N/D <ANS THENA$="TOO SMALL":PROC_ERROR:GOTO510
800 IF N/D =ANS THENA$="THE SAME":PROC_ERROR:GOTO510
810 IF N/D >=1 THENA$="BELOW 1 PLEASE":PROC_ERROR:GOTO510
820 H=H+1:F$=NF$
830 FORT=255 TO 0 STEP-15
840 SOUND 1,1,T,2
850 NEXTT
860 ANS=N/D
870 ENDPROC
880 DEFPROC_ERROR

```

```

890 REM      DISPLAY VALIDATION ERROR
900 SOUND 1,2,50,20
910 PRINT TAB(16,16);CHR$(133);CHR$(136);CHR$(141);A$
920 PRINT TAB(16,17);CHR$(133);CHR$(136);CHR$(141);A$
930 ENDPROC
940 DEFPROC_TITLE
950 REM      DISPLAY TITLE OF PROGRAM
960 TITLE$="FRACTION TOWERS"
970 PRINT TAB(7,5);CHR$(131);CHR$(141);TITLE$
980 PRINT TAB(7,6);CHR$(131);CHR$(141);TITLE$
990 FOR T=1TO250STEP15
1000     SOUND 1,-15,T,1
1010 NEXT
1020 PRINT TAB(4,11);CHR$(130);CHR$(136);"PRESS ANY KEY TO
START"
1030 IF INKEY$(0)=" " THEN 1030
1040 REM      DISPLAY INSTRUCTIONS
1050 CLS
1060 PRINT TAB(0,4);CHR$(134);
1070 REM SEND CHARS TO WRITING SUB
1080 A$="YOU ARE GOING TO BUILD FRACTION TOWERS"
1090 PROC_writing
1100 PRINT TAB(0,7);CHR$(130);
1110 A$="EACH FRACTION THAT YOU CHOOSE MUST BE"
1120 PROC_writing
1130 PRINT TAB(0,10);CHR$(129);
1140 A$="BIGGER THAN THE ONE BEFORE AND MUST BE"
1150 PROC_writing
1160 PRINT TAB(12,13);CHR$(131);
1170 A$="LESS THAN 1"
1180 PROC_writing
1190 PRINT TAB(3,19);CHR$(135);CHR$(136);"PRESS ANY KEY TO
CONTINUE"
1200 IF INKEY$(0)=" " THEN 1200
1210 ENDPROC
1220 REM USER HAS PRESSED 'ESCAPE'
1230 MODE7
1240 PRINT TAB(5,10);CHR$(129);"YOU PRESSED THE ESCAPE KEY!
!!"
1250 PRINT TAB(5,12);"DOES THIS MEAN YOU DON'T LIKE THE"
1260 PRINT TAB(5,13);"PROGRAM (Y/N)"
1270 YN$=GET$
1280 IF LEFT$(YN$,1)="N" THEN RETURN
1290 FOR T=255TO0 STEP-1
1300     SOUND 1,-15,T,0
1310 NEXT T
1320 PRINT TAB(10,15);CHR$(130);CHR$(141);"BYE"
1330 PRINT TAB(10,16);CHR$(130);CHR$(141);"BYE"
1340 END
1350 DEFPROC_writing
1360 REM PROCEDURE TO PUT WRITING ON SCREEN
1370 FOR T=1TO LEN(A$):PRINT MID$(A$,T,1);
1380     SOUND 1,-10,(255/LEN(A$))*T,1
1390     FOR U=1TO 100:NEXT U
1400 NEXT T
1410 ENDPROC
1420 REM DATA FOR FRACTIONS
1430 DATA "1/10",.1,"1/4",.25,"4/5",.8
1440 DATA END,0
1450 DEFPROC_FINISH
1460 REM COMPLETED ALL FRACTIONS
1470 CLS
1480 PRINT"" YOU HAVE DONE VERY WELL"
1490 PRINT"" THANK YOU FOR PLAYING"
1500 SOUND 1,1,10,10:SOUND 1,2,10,10
1510 END

```