
3. Boating

General Description

The object of this next simulation is to compel the user to
consider logical problem solving techniques in action in a (rather
unlikely) real life simulation.

User instructions appear on the screen as soon as the program
is run. The object of the exercise is to transport two men and two
girls across a river. But there's a problem. The only available
boat has a very limited capacity: at any one time it can only
carry a maximum of one man and two girls. What is the
minimum number of times that the boat has to cross the river
before all four are on the other side?

Detailed Description

Lines 1-2 Is the first part of the 'Delete' trick.
10 Check for Escape key.
15-80 Sets up variables and sound.
90-110 Sets up colours.
120-160 Draw river
170-220 GOTO procedures.
210 Puts the boat on the opposite side of the river.
230-320 Displays G's and M's, also displays number of rides

taken.
330-380 Check to see if everybody is on the same side of the

river but not on the same side as where they started, if so, then
go to finish.

390-430 Routine for finish
440-650 Get input from keyboard e.g. G, GG, or M, and

validate. Also move person across river in memory but not
visually.

660 Data for validation of keyboard input.
670-870 Move boat across river visually.
880-920 Escape key has been pressed.
1000-1130 Display instructions.
1140-1188 Second part of the delete trick.

Educational Notes

It is often difficult to demonstrate to students that logic has a
place ion the real world, and that many problems are not most
effectively dealt with by means of intuition or commonsense.
Textbook examples often tend to reinforce the suspicion that
logic is mathematics in the esoteric mode. This program has
been designed to breathe a little life into logical examples.

Boating will run on both a Model A & B, and as the listing
notes, it would be fatal to renumber this program.

Program Listing

>
    1 ONERRORGOTO10
    2 GOTO1000
   10 ON ERROR GOSUB 880
   15 CLEAR
   20 C=0:NR=0
   30 ENVELOPE 1,1,1,-1,0,20,20,50,124,-3,-1,-1,126,126
   40 S$="L":OS$="R"
   50 DIM P$(4)
   60 FORT=1TO4
   70   P$(T)="L"
   80 NEXT
   90 MODE5:VDU23;8202;0;0;0;
  100 VDU 19,0,2,0,0,0
  110 VDU 19,1,6,0,0,0
  120 GCOL 0,1
  130 MOVE 0,600
  140 MOVE 0,800
  150 PLOT 85,1279,600
  160 PLOT 85,1279,800
  170 PROC_DISPLAY
  180 PROC_KEYBOARD
  190 PROC_BOAT
  200 PROC_CHECK_FINISH
  210 D$=S$:S$=OS$:OS$=D$:NR=NR+1
  220 GOTO 90
  230 DEF PROC_DISPLAY
  240 COLOUR 2
  250 FOR T=1 TO 4
  260   IF P$(T)="L"ANDT<3THENPRINT TAB(10+T,5);"G"
  270   IF P$(T)="R"ANDT<3THENPRINT TAB(10+T,14);"G"
  280   IF P$(T)="L"ANDT>2THENPRINT TAB(10+T,5);"M"
  290   IF P$(T)="R"ANDT>2THENPRINT TAB(10+T,14);"M"

  300 NEXT
  310 PRINT TAB(0,1);"NUMBER OF RIDES=";NR
  320 ENDPROC
  330 DEF PROC_CHECK_FINISH
  340 C=0:FORT=1TO4
  350   IF P$(T)="R"C=C+1
  360 NEXTT
  370 IF C=4 PROC_FINISH
  380 ENDPROC
  390 DEF PROC_FINISH
  400 CLG
  410 COLOUR3
  420 PRINTTAB(1,10);"Well done"'"You crosses the"'"river ";
NR+1;" times"
  430 END
  440 DEF PROC_KEYBOARD
  450 PRINT TAB(3,20);
  460 INPUT"MOVE >" M$
  470 L=LEN(M$)
  480 IF L=0 OR L>2 THEN 540
  490 RESTORE
  500 FOR T=1TO3
  510   READ D$
  520 IF D$=M$THENE=T:T=3:NEXT:GOTO570
  530 NEXT
  540 PRINT TAB(9,20);SPC(L)
  550 VDU 7
  560 GOTO 450
  570 FOR Q=1TO4
  580   T=E
  590   IF P$(Q)<>S$ THEN 630
  600 IF T=1 AND Q<3 THEN P$(Q)=OS$:Q=4:NEXT:GOTO650
  610 IF T=2 AND Q>2 THEN P$(Q)=OS$:Q=4:NEXT:GOTO650
  620 IF T=3 AND P$(1)+P$(2)=S$+S$ THEN P$(1)=OS$:P$(2)=OS$:
Q=4:NEXT:GOTO650
  630 NEXTQ
  640 GOTO 540
  650 ENDPROC
  660 DATA G,M,GG,
  670 DEF PROC_BOAT
  680 IF S$="R"THEN 780
  690 FOR Y=7TO 12
  700   COLOUR 133
  710   PRINT TAB(12,Y);"#"
  720   SOUND 1,-15,123,1
  730   A$=INKEY$(5)
  740   COLOUR 129
  750   PRINT TAB(12,Y);" "
  760 NEXT
  770 GOTO 860
  780 FOR Y=12 TO 7 STEP-1
  790   COLOUR 133
  800   PRINT TAB(12,Y);"#"
  810   SOUND 1,-15,123,1
  820   A$=INKEY$(5)
  830   COLOUR 129
  840   PRINT TAB(12,Y);" "
  850 NEXT
  860 SOUND 1,1,100,20
  870 ENDPROC
  880 MODE7
  890 PRINTTAB(10,10)"Do you want to quit (Y/N)";
  900 A$=GET$
  910 IF A$="Y"THENMODE7:END
  920 RETURN
 1000 MODE7
 1010 PRINTTAB(10,3)"BOATING."
 1020 PRINT'''" The idea of this game is to get the"
 1030 PRINT"two girls and two men across the river"

 1040 PRINT"in a small boat."
 1050 PRINT'"The boat can carry a maximum of 2 girls"
 1060 PRINT"or 1 man."
 1070 PRINT'"So at any time the boat can carry 1 man,";
 1080 PRINT"1 girl or 2 girls." 
 1090 PRINT'"I think the minimum number of crosses is";
 1095 PRINT"9 - see if you can beat it."
 1097 PRINT'"Type: 'G' for 1 girl ,'GG' for 2 girls,"
 1098 PRINT"and 'M' for 1 man."
 1100 PRINT"Do you wish to stop the program (Y/N)";
 1110 A$=GET$
 1120 IFA$="Y"THENEND
 1130 PRINTA$
 1140 PRINT''"Press Function Key 'f0'"
 1150 *KEY0DELETE 1000,1200|MRUN|M
 1160 IFINKEY(-33)=0THEN1160
 1170 END
 1180 *******************************
 1181 *         BOATING             *
 1182 *        WRITTEN BY           *
 1183 *     Andrew Pusey. FEB 1983  *
 1184 *******************************
 1185            NOTE
 1186   DO NOT RENUMBER THIS PROGRAM
 1187  
 1188  

