
9. Contact-4

General Description

This is a variation on noughts and crosses for two players. Player
one has red blocks, player two yellow. When the program is run
a grid appears on the screen, and the idea is to drop one of your
blocks into the grid using the cursor control keys to move, and
the Return key to drop the block. The object of the game is for a
player to produce a line of four blocks in any direction -
vertically, horizontally or diagonally.

Detailed Description

Lines 1-2 First part of delete block.
10-50 Set up variables and sound.
60 Disable cursor control keys.
70-120 Draw grid.
130-140 Go to procedures to get input from keyboard and

check for line.
150-210 If line has been made then say who has won.
220-280 Procedure to display block on screen.

CL = colour of block.
290-320 Procedure to get input from keyboard.

If player 1 has just moved then new player = player 2 and vice-
versa. Put block at starting position.

330-420 Display player on screen and get input from
keyboard.

430-490 Move block across screen.
500-620 Drop block down the grid and store the position

where it rests in B(r,c) r = row c = column.
End procedure.

630-1000 Procedure to check for a line.
Line 640-700 check for horizontal line.
Line 710-770 check for vertical line.
Line 780-870 check for / line.
Line 880-960 check for \ line.
Line 970-990 check to see if top column is full, if so then set E =
draw.

1010-1210 Instructions.
1220-1250 Second part of delete trick.
1260-1300 Credits.

Educational Notes

Whilst this program has a certain value as a logic exercise, it has
been designed primarily as entertainment.

Program Listing

>
    1 ON ERRORGOTO10
    2 GOTO1010
   10 CL=2
   20 E=0
   30 ENVELOPE 1,1,10,-10,10,20,40,20,126,0,0,0,126,126    
   40 MODE5:VDU23;8202;0;0;0;
   50 DIM B(7,7)
   60 *FX4,1
   70 FOR T=0 TO 8
   80   MOVE 100,100+(T*80)
   90   DRAW 900,100+(T*80)  
  100   MOVE 100+(T*100),100
  110   DRAW 100+(T*100),740
  120 NEXT
  130 PROC_GO
  140 PROC_CHECK
  150 IF E=00THEN130
  160 PRINTTAB(0,0);SPC(120);TAB(0,1)
  170 IF E=3PRINT"IT'S A DRAW.":END
  180 PRINT "PLAYER ";E;" HAS WON."
  190 SOUND1,1,1,30
  200 SOUND1,0,0,0
  210 END
  220 DEFPROC_PUT_BLOCK
  230 MOVE 108+R*100,108+C*80
  240 MOVE 108+R*100,92+(C+1)*80 
  250 GCOL0,CL
  260 PLOT 85,92+(R+1)*100,108+C*80
  270 PLOT 85,92+(R+1)*100,92+(C+1)*80 
  280 ENDPROC
  290 DEFPROC_GO
  300 IF CL=1CL=2ELSECL=1
  310 C=8:R=4
  320 PROC_PUT_BLOCK

  330 CF=CL
  340 PRINTTAB(0,0);SPC(120);TAB(0,1);
  350 PRINT "Player ";CL
  360 A=GET
  370 D=0
  380 IFA=137D=1
  390 IFA=136D=-1
  400 IFA=13D=0:GOTO420
  410 IFD=0VDU7:GOTO360
  420 CL=0
  430 PROC_PUT_BLOCK
  440 CL=CF
  450 R=R+D
  460 IFR=8R=0
  470 IFR=-1R=7
  480 IFD<>0THEN320
  490 IF B(R,7)<>0VDU7:GOTO320
  500 FOR C=7 TO0 STEP-1
  510   IF C=0 P=0:GOTO580
  520   IF B(R,C-1)<>0P=C:C=0:GOTO580
  530   PROC_PUT_BLOCK
  540   SOUND 1,-15,(C+1)*30,1
  550   CL=0
  560   PROC_PUT_BLOCK
  570   CL=CF
  580 NEXT
  590 CL=CF:C=P
  600 PROC_PUT_BLOCK
  610 B(R,P)=CL
  620 ENDPROC
  630 DEFPROC_CHECK
  640 REM CHECK HORIZONTAL
  650 FORT=0TO4
  660   Z=B(T,C)
  670   IFZ=0THEN690
  680   IFZ=B(T+1,C)ANDZ=B(T+2,C)ANDZ=B(T+3,C)THENT=10
  690 NEXTT
  700 IFT=11 E=Z:ENDPROC
  710 REM VERTICALLY
  720 FORT=0TO4
  730   Z=B(R,T)
  740   IFZ=0THEN760
  750   IF Z=B(R,T+1)ANDZ=B(R,T+2)ANDZ=B(R,T+3)THENT=10
  760 NEXTT
  770 IFT=11THENE=Z:ENDPROC
  780 REM DIAGONAL /
  790 F=R:G=C
  800 IFR>0ANDC>0THENR=R-1:C=C-1:GOTO800
  810 FORT=0TO4
  820   IFR+T>4ORC+T>4THENT=4:GOTO860
  830   Z=B(R+T,C+T)
  840   IFZ=0THEN860
  850   IF Z=B(R+T+1,C+T+1)ANDZ=B(R+T+2,C+T+2)ANDZ=B(R+T+3,C
+T+3)THENT=10
  860 NEXTT
  870 IFT=11THENE=Z:ENDPROC
  880 REM DIAGGONAL
  890 IFF<7ANDG>0THENF=F+1:G=G-1:GOTO890
  900 FORT=0TO4
  910   IFF-T<3ORG+T>4THENT=4:GOTO950
  920   Z=B(F-T,G+T)
  930   IFZ=0THEN950
  940   IF Z=B(F-T-1,G+T+1)ANDZ=B(F-T-2,G+T+2)ANDZ=B(F-T-3,G
+T+3)THENT=10
  950 NEXTT
  960 IFT=11THENE=Z
  970 REM IS IT A DRAW ?
  980 I=0:FORT=0TO7:IFB(T,7)<>0THENI=I+1
  990 NEXT:IFI=8THENE=3

 1000 ENDPROC
 1010 MODE7
 1020 TITLE$=CHR$(141)+CHR$(133)+"CONNECT 4"
 1030 PRINTTAB(10,5);TITLE$';TAB(10);TITLE$
 1040 PRINTTAB(3,10);CHR$136"Press any key for instructions.
"
 1050 A$=GET$
 1060 CLS:VDU7 
 1070 PRINTTAB(1,3);
 1080 PRINT"This game is a bit like naughts and"
 1090 PRINT"crosses."
 1100 PRINT'"This is a two player game; player 1 has"
 1110 PRINT"red blocks, and player 2 has yellow"
 1120 PRINT"blocks."
 1130 PRINT'"The idea is, that you drop your blocks"
 1140 PRINT"down the grid, and if you can make a"
 1150 PRINT"line of 4 of your colour, then you win."
 1160 PRINT'"You can get lines horizontal, vertical"
 1170 PRINT"or diagonal."
 1180 PRINT'"To move press the cursor control keys"
 1190 PRINT"   Left [    ] Right"
 1200 PRINT'"To drop your block, press the RETURN"
 1210 PRINT"key."
 1220 PRINT'"Press the Function Key 'f0'"
 1230 *KEY0DELETE1010,1300|MRUN|M
 1240 IFINKEY(-33)=FALSE THEN1240
 1250 END
 1260 ******************************
 1265 *         CONNECT 4          *
 1270 *        WRITTEN BY          *
 1275 *    Andrew Pusey. FEB 1983  *
 1280 ******************************
 1285  
 1290   DO NOT RENUMBER THIS PROGRAM
 1295  
 1300  

