

20. Four Stroke Engine

General Description

This program is a graphics demonstration of how a four stroke engine works. It will run on a Model 'A' and 'B'. Once running, the program continues until you press 'ESC'. It only takes a few seconds to load and is great to watch. To develop the program, you may wish to add labels to the spark plugs, valves, piston etc, and a facility to freeze the display - in which case you may run out of memory on a Model 'A', but as it stands the valves open and shut, and airflow is indicated with arrows.

Detailed Description

Lines 10-160 Main structure: the UNTIL FALSE guarantees perpetual motion for the engine.

170-510 The effective speed of the demonstration can be controlled by making the value of 'step' smaller in line 320. This in turn varies the number of 'frames' into the POSITION array.

520-630 In line 570 notice the exclusive OR of the foreground graphical colour to update the graphical display.

630-1010 Routines to move the piston.

1020-end Draws the cylinder and provides the spark.

Educational Notes

This is a good substitute for 'chalk and talk'. If you decide not to label the various parts of the engine then you can use the program to elicit responses as to what is happening from the class.

Program Listing

```

10 REM =====
20 REM   4 STROKE ENGINE
30 REM
40 REM   (C)   B. CROW. 1983
50 REM   .....
60 MODE5:VDU23;8202;0;0;0;
70 PROCinit
80 REPEAT
90   FORI%=0TOframes%-1
100    PROCplot(1+J%,1)
110    IF I%=tdc% OR I%= bdc% THEN PROCstroke
120    PROCnewframe(1+J%)
130    PROCplot(2-J%,0)
140    PROCnextposition
150  NEXT
160 UNTIL FALSE
170 REM =====
180 REM   PROCinit
190 REM   .....
200 REM
210 DEF PROCinit
220 PRINT" THE 4 STROKE ENGINE"
230 VDU 23;8202;0;0;0;
240 VDU 19,1,0,0,0,0
250 VDU 19,2,0,0,0,0
260 VDU 23,240,8,8,28,28,28,8,8,8
270 PRINTTAB(14,6);CHR$(240)
280 VDU 29,930;245;
290 VDU28,0,15,11,14
300 PROCcylinder
310 S%=1
320 step=0.261
330 frames%=2*PI/step
340 C%=140
350 R%=350
360 tdc%=0
370 bdc%=frames% DIV 2
380 DIM POSITION%(frames%)
390 FORI%=0TOframes%-1
400   POSITION%(I%)=C%*SIN(I%+step+step*0.5)
410 NEXT
420 POSITION%(I%)=POSITION%(0)
430 DIM stroke$(3)
440 stroke$(0)=" INDUCTION  "
450 stroke$(1)="COMPRESSION"
460 stroke$(2)="   POWER   "
470 stroke$(3)=" EXHAUST   "
480 J%=1
490 I%=0
500 PROCplot(2-J%,0)
520 REM =====
530 REM   PROCplot
540 REM   .....
550 REM
560 DEF PROCplot(colour%,pos%)
570 GCOL 3,colour%
580 PLOT4,0,0
590 PLOT5,POSITION%(I%+pos%),POSITION%((I%+pos%+6) MODframes%)
600 PLOT5,0,R%+POSITION%((I%+pos%+6) MODframes%)
610 PLOT0,-75,-30:PLOT1,0,60:PLOT1,150,0:PLOT1,0,-60
620 ENDPROC
630 REM =====
640 REM   PROCnewframe

```

```

650 REM .....
660 REM
670 DEF PROCnewframe(colour%)
680 VDU 19,colour%,7,0,0,0
690 VDU 19,3-colour%,0,0,0,0
700 ENDPROC
710 DEF PROCnextposition
720 J%=1-J%
730 ENDPROC
740 DEF PROCstroke
750 S%=(S%+1) MOD 4
760 PRINT stroke$(S%)
770 GCOL3,3
780 IF S%=0 OR S%=3PROCoutvalve
790 IF S%=0 OR S%=1PROCinvalve
800 IF S%=2 OR S%=3PROCspark
810 ENDPROC
820 REM =====
830 REM PROCinvalve
840 REM .....
850 REM
860 DEF PROCinvalve
870 PLOT 4,-70,570
880 PLOT 1,30,0
890 PLOT 0,0,-20
900 PLOT 1,-30,0
910 ENDPROC
920 REM =====
930 REM PROCoutvalve
940 REM .....
950 REM
960 DEF PROCoutvalve
970 PLOT 4,70,570
980 PLOT 1,-30,0
990 PLOT 0,0,-20
1000 PLOT 1,30,0
1010 ENDPROC
1020 REM =====
1030 REM PROCcylinder
1040 REM .....
1050 REM
1060 DEF PROCcylinder
1070 PLOT 4,-85,210
1080 PLOT1,0,360
1090 PLOT1,170,0
1100 PLOT1,0,-360
1110 PLOT4,-70,570
1120 PLOT1,0,30
1130 PLOT1,-90,0
1140 PLOT0,-100,15
1150 PLOT1,90,0
1160 PLOT1,-15,-15
1170 PLOT81,0,30
1180 PLOT0,25,0
1190 PLOT1,120,0
1200 PLOT1,0,-60
1210 PLOT1,80,0
1220 PLOT1,0,60
1230 PLOT1,120,0
1240 PLOT0,10,-10
1250 PLOT1,90,0
1260 PLOT1,-15,15
1270 PLOT81,0,-30
1280 PLOT4,70,570
1290 PLOT1,0,30
1300 PLOT1,90,0
1310 ENDPROC
1320 REM =====
1330 REM PROCspark

```

```
1340 REM .....
1350 REM
1360 DEF PROCspark
1370 PLOT4,0,550
1380 PLOT1,-15,-15
1390 PLOT1,30,0
1400 PLOT1,-15,-15
1410 ENDPROC
```