

28. Simon

General Description

This program is a computer version of the SIMON memory game available in High Street shops. I have included it because I like the way Ian Clarke has handled the sound and display and I believe there is a place in any suite of programs for a memory tester such as this.

When you run the program you will see four coloured bars grouped in a square shape. There are numbers next to each bar (1-4), and these are the keys that control any bar - key 1 controls bar 1.

If you are not familiar with SIMON style games, then read on. The computer makes a sound and flashes a bar. You must then repeat exactly what the computer has generated by pressing the key corresponding to the bar that flashed. To begin with the sounds and flashes are simple but quickly become more difficult.

The computer will tell you how many goes you survived. This runs on a Model 'A'.

Detailed Description

Lines 10-300 This sets level of play. Auto repeat and cursor are turned off. The S() array holds the time gaps for each level of play before the next sound and bar colour flash generated. Array T() holds the number of goes before the game is considered even at that level. Invalid characters generate a 'beep' and the valid characters are converted to numbers 1-5 (line 300).

310-440 These print the colour bars on the screen.

450-620 This is the actual game. The subroutines in line 510 flash the bar. The array G() holds the computer's random sound and bar flashes. The computer checks each response as you GET

it in from the keyboard.

630-690 The success routine.

700-1090 These are the sound and flash routines. Line 970 sets the colour pitch for the flash colours, bars 1 and 3. Line 750 does the same for bars 2 and 4. The sound value 5 is called in 980 and 760 respectively.

1100-1230 The fail routine and the option for another go. Line 1190 could use the more sophisticated validation shown elsewhere.

Educational Notes

Young children in particular need assistance in developing retentive facilities and I think SIMON style games manage to do this. It avoids cramming useless facts into the mind, merely developing the ability to remember. I have found that first years enjoy it as an occasional end of lesson filler. It can hold the attention of a small group of youngsters for about 10 minutes.

Program Listing

```
10 REM *****
20 REM **      SIMON      **
30 REM **      Written by Ian Clarke.**
40 REM **      Jan 83      **
50 REM *****
60 ON ERROR GOTO 1200
70 ENVELOPE 1,2,4,-8,4,8,16,8,1,0,0,-2,120,120
80 *FX4,1
90 *FX11,0
100 MODE 7
101 VDU 23;8202;0;0;0;
102 PRINT
110 DIM S(5),T(5),S$(2)
120 S$(0)="es."
130 S$(1)=". "
140 S(1)=800:T(1)=10
150 S(2)=500:T(2)=15
160 S(3)=300:T(3)=20
170 S(4)=150:T(4)=25
180 S(5)=50 :T(5)=30
190 PRINT TAB(3,6);CHR$(141);CHR$(131);"What level do you
want to play at?"
200 PRINT TAB(3);CHR$(141);CHR$(131);"What level do you w
ant to play at?"
210 PRINT TAB(10,10);CHR$(141);CHR$(131);"1,2,3,4 OR 5."
220 PRINT TAB(10);CHR$(141);CHR$(131);"1,2,3,4 OR 5."
230 FOR X=1 TO 200:NEXT
240 *FX15
```

```

250 A=GET
260 IF A<49 THEN VDU 7:GOTO 230
270 IF A>53 THEN VDU 7:GOTO 230
280 PRINT TAB(15,13);CHR$(141);CHR$(131);A-48
290 PRINT TAB(15);CHR$(141);CHR$(131);A-48
300 A=A-48
310 DIM G(T(A))
320 FOR X=1 TO 1000:NEXT
330 CLS
340 D=0
350 GOSUB 750
360 PRINT TAB(18,4);CHR$(133);"1"
370 GOSUB 970
380 PRINT TAB(5,12);CHR$(129);"2"
390 D=5:
400 GOSUB 750
410 PRINT TAB(18,22);CHR$(132);"3"
420 D=52
430 GOSUB 970
440 PRINT TAB(32,12);"4"
450 FOR X=1 TO 1000:NEXT
460 FOR H=1 TO T(A)
470 G(H)=RND(4)
480 FOR X=1 TO S(A):NEXT
490 FOR J=1 TO H
500 D=G(J)+48
510 ON G(J) GOSUB 750,970,750,970
520 NEXT J
530 FOR J=1 TO H
540 FOR X=1 TO 100:NEXT
550 D=GET
560 IF D<49 THEN VDU 7:GOTO 540
570 IF D>53 THEN VDU 7:GOTO 540
580 IF D-48<>G(J) THEN GOTO 1100
590 ON D-48 GOSUB 750,970,750,970
600 NEXT J
610 FOR X=1 TO 1000:NEXT
620 NEXT H
630 CLS
640 FOR X=1 TO 1000:NEXT
650 PRINT TAB(5,8);CHR$(141);CHR$(131);"Congratulations o
n completing"
660 PRINT TAB(5);CHR$(141);CHR$(131);"Congratulations on
completing"
670 PRINT TAB(13);CHR$(141);CHR$(131);"level ";A;". "
680 PRINT TAB(13);CHR$(141);CHR$(131);"level ";A;". "
690 GOTO 1140
700 REM -----
710 REM This subroutine produces
720 REM sound for the top bar or
730 REM bottom bar and flashes
740 REM the bar off then on.
750 IF D=51 THEN S=81:Y1=21:C=132 ELSE S=53:Y1=5:C=133
760 SOUND 1,-15,S,10
770 FOR X=0 TO 20
780 D$=D$+CHR$(32)
790 NEXT
800 FOR X=1 TO S(A)/4:NEXT
810 PRINT TAB(8,Y1);D$
820 D$=""
830 FOR X=1 TO 20
840 D$=D$+CHR$(255)
850 NEXT
860 PRINT TAB(8,Y1);CHR$(C);D$
870 D$=""
880 FOR X=1 TO S(A):NEXT
890 SOUND &0011,0,0,0
900 REM .....
910 RETURN

```

```

920 REM -----
930 REM This subroutine produces
940 REM sound for the left bar or
950 REM right bar and flashes
960 REM the bar off then on.
970 IF D=52 THEN S=101:X1=29:C=130 ELSE S=69:X1=7:C=129
980 SOUND 1,-15,S,10
990 FOR Y=6 TO 20
1000 PRINT TAB(X1+1,Y);CHR$(32)
1010 NEXT
1020 FOR X=1 TO S(A)/4:NEXT
1030 FOR Y=6 TO 20
1040 PRINT TAB(X1,Y);CHR$(C);CHR$(255)
1050 NEXT
1060 FOR X=1 TO S(A):NEXT
1070 SOUND &0011,0,0,0
1080 REM .....
1090 RETURN
1100 CLS
1110 IF H=2 THEN W=1 ELSE W=0
1120 PRINT TAB(5,10);CHR$(141);CHR$(131);"You failed after
";H-1;" go";S$(W)
1130 PRINT TAB(5);CHR$(141);CHR$(131);"You failed after ";
H-1;" go";S$(W)
1140 FOR X=1 TO 5000:NEXT
1150 PRINT TAB(5,13);CHR$(141);CHR$(131);"Do you want anot
her game?"
1160 PRINT TAB(5);CHR$(141);CHR$(131);"Do you want another
game?"
1170 *FX15
1180 D$=GET$
1190 IF D$="Y" OR D$="y" THEN RUN
1200 SOUND 1,1,120,70
1210 CLS
1220 PRINT TAB(15,12);CHR$(136);CHR$(141);CHR$(134);"BYE!
!"
1230 PRINT TAB(15);CHR$(136);CHR$(141);CHR$(134);"BYE!!!"
1240 FOR X=1 TO 8500:NEXT
1250 MODE 7
1260 *FX4
1270 *FX11,25
1280 END

```