

30. Guessnum

General Description

There are several versions of this kind of program available. This program displays yours and the computer's responses nicely in Mode 7. It will run on a BBC Model 'A'. This program is the only one in the book that actually asks your name, a nice touch I think. It made me see that, given a data base of names, the computer could generate a 'name guess' as part of its interaction with the user. The computer generates some rather rude messages for 'stupid' guesses, and keeps track of the number of wasted goes.

Detailed Description

Lines 10-470 This is the main structure of the program. The cursor and auto repeat are disconnected. The lines 410-460 play the game. The flag at 440 decides whether or not you give advice on the number entered being below or above the computer's number.

480-700 Carriage return exits from the validation which only accepts digits 0-9 and two characters only.

710-990 The rude messages procedure. The computer keeps track of your new entry and its previous advice and if you do not respond sensibly, will abuse you.

1000-1150 General advice routine. This routine is only entered if the flag at line 960 has not been reset.

1160-end Nothing surprisingly here. Lines 1210-1260 write in double size characters by looping whole lines. 1280-1310 do it more traditionally. The flag is set back to zero before playing the game again. All editing facilities are restored before exit.

Educational Notes

At one level this program is merely a game, but at the other it is useful for teaching a binary search. As such it is quite a valuable aid for teaching fifth and sixth year computer studies youngsters. A binary search presupposes data that is kept in some form of order, and obviously numbers 0 to 100 are kept in numerical order. If you are not familiar with the binary search, consider the digits 0-9 below. (The computer has thought of 4)

0 1 2 3 4 5 6 7 8 9

Ask ' is the sought number in the top half of the digits' ?

When you know the answer to that you repeat the question for the digits within which range you know the number lies. So the number ' 4 lies in the bottom half of the numbers . . . ask the question again and you learn that ' 4 must lie between 3 and 4, and you are virtually there. For the more able you can discuss retrieval times based on the number of elements of initial data.

Program Listing

```
10 REM *****
20 REM **      GUESS MY NUMBER      **
30 REM **      Written by Ian Clarke.**
40 REM **      Jan 83      **
50 REM *****
60 *FX11,0
70 ON ERROR GOTO 1460
80 ENVELOPE 1,2,2,3,4,4,4,4,1,-1,0,-1,127,70
90 ENVELOPE 2,2,-2,-3,-4,4,4,4,1,-1,0,-1,127,70
100 ENVELOPE 3,2,3,-1,2,3,4,2,2,-1,0,-1,100,80
110 ENVELOPE 4,1,0,0,0,5,5,5,1,0,0,-1,127,127
120 DIM OE$(2)
130 OE$(0)="go."
140 OE$(1)="goes."
150 N=RND(101)-1
160 FLAG=0
170 B=-1
180 A=101
190 T1=0
200 MODE 7
210 VDU 23;8202;0;0;0;
220 PRINT
230 *FX4,1
240 PRINT TAB(10,7);"What is your name?"
250 PRINT TAB(10);
260 INPUT N$
270 CLS
280 PRINT TAB(3,2);CHR$(141);CHR$(134);"HELLO ";N$
290 PRINT TAB(3);CHR$(141);CHR$(134);"HELLO ";N$
300 FOR X=1 TO 5000:NEXT
310 FOR X=0 TO 1
```

```

320     PRINT TAB(5,6+X);CHR$(141);CHR$(130);"I am thinking
of a whole number"
330     PRINT TAB(5,8+X);CHR$(141);CHR$(130);"between 0 and
100."
340     PRINT TAB(5,10+X);CHR$(141);CHR$(130);"You have to
guess the number"
350     PRINT TAB(5,12+X);CHR$(141);CHR$(130);"in as few gu
esses as possible."
360     NEXT X
370     PRINT TAB(10,22);CHR$(130);"Press a key to start."
380     *FX15
390     D=GET
400     CLS
410     PROCgetguess
420     IF G=N THEN GOTO 1170
430     PROCerrorcheck
440     IF FLAG=-1 THEN FLAG=NOT FLAG:GOTO 410
450     PROCbelabo
460     GOTO 400
470     END
480     REM -----
490     REM         Get user's guess
500     REM         and verify.
510     DEFPROCgetguess

520     T=T+1
530     G$=""
540     FOR X=0 TO 1
550         PRINT TAB(5,2+X);CHR$(141);CHR$(131);"Enter your gu
ess."
560     NEXT
570     FOR X=1 TO 100:NEXT
580     *FX15
590     D=GET
600     IF D=13 THEN GOTO 680
610     IF D<48 THEN VDU 7:GOTO 570
620     IF D>57 THEN VDU 7:GOTO 570
630     IF LEN(G$)=3 THEN VDU7:GOTO 570
640     G$=G$+CHR$(D)
650     PRINT TAB(10,6);CHR$(141);CHR$(131);G$
660     PRINT TAB(10);CHR$(141);CHR$(131);G$
670     GOTO 570
680     G=VAL(G$)
690     REM .....
700     ENDPROC
710     REM -----
720     REM         validate answer
730     DEFPROCerrorcheck
740     IF G<101 THEN GOTO 810
750     T1=T1+1
760     SOUND 0,4,5,50
770     FOR X=0 TO 1
780         PRINT TAB(5,15+X);CHR$(141);CHR$(133);"The number i
s between 1 and 100."
790     NEXT
800     GOTO 940
810     IF G>B THEN GOTO 880
820     SOUND 0,4,5,50
830     FOR X=0 TO 1
840         PRINT TAB(5,15+X);CHR$(141);CHR$(133);"you nit twit
, I told you the"
850         PRINT TAB(5,17+X);CHR$(141);CHR$(133);"number was g
reater than ";B;". "
860     NEXT X
870     GOTO 940
880     IF G<A THEN GOTO 970
890     SOUND 0,4,5,50
900     FOR X=0 TO 1

```

```

 910 PRINT TAB(5,15+X);CHR$(141);CHR$(133);"you nit twit
, I told you the"
 920 PRINT TAB(5,17+X);CHR$(141);CHR$(133);"number was 1
ess than ";A;". "
 930 NEXT X
 940 FOR X=1 TO 5000:NEXT
 950 T1=T1+1
 960 FLAG=NOTFLAG
 970 CLS
 980 REM .....
 990 ENDPROC
1000 REM -----
1010 REM Give the user a clue
1020 REM to the answer
1030 DEFPROCbelabo
1040 IF G>N THEN GOTO 1100
1050 SOUND 1,1,120,40
1060 PRINT TAB(5,15);CHR$(141);CHR$(134);"The number is gr
eater than ";G
1070 PRINT TAB(5);CHR$(141);CHR$(134);"The number is great
er than ";G
1080 B=G
1090 GOTO 1140
1100 SOUND 1,2,120,40
1110 PRINT TAB(5,15);CHR$(141);CHR$(129);"The number is le
ss than ";G
1120 PRINT TAB(5);CHR$(141);CHR$(129);"The number is less
than ";G
1130 A=G
1140 FOR X=1 TO 7000:NEXT
1150 REM .....
1160 ENDPROC
1170 ENVELOPE 4,2,3,-6,3,10,20,30,1,0,0,-1,120,120
1180 SOUND 1,4,120,40
1190 CLS
1200 FOR X=0 TO 1
1210 IF T=1 THEN A=0 ELSE A=1
1220 PRINT TAB(4,13+X);CHR$(141);CHR$(133); " You got t
he answer in ";T;" ";OE$(A)
1230 IF T1=0 THEN GOTO 1260
1240 IF T1=1 THEN A=0 ELSE A=1
1250 PRINT TAB(5,15+X);CHR$(141);CHR$(133);"But you wast
ed ";T1;" "OE$(A)
1260 NEXT X
1270 FOR X=1 TO 7000:NEXT
1280 PRINT TAB(12);CHR$(141);CHR$(133);"Do you want anothe
r"
1290 PRINT TAB(12);CHR$(141);CHR$(133);"Do you want anothe
r"
1300 PRINT TAB(12);CHR$(141);CHR$(133);"go ";N$
1310 PRINT TAB(12);CHR$(141);CHR$(133);"go ";N$
1320 *FX15
1330 YN$=GET$
1340 PRINT TAB(20,21);CHR$(141);YN$
1350 PRINT TAB(20);CHR$(141);YN$
1360 FOR X=1 TO 1000:NEXT
1370 IF YN$<>"Y" THEN 1460
1380 ENVELOPE 4,1,0,0,0,5,5,5,1,0,0,-1,127,127
1390 N=RND(101)-1
1400 B=-1
1410 A=101
1420 FLAG=0
1430 T=0
1440 T1=0
1450 GOTO 400
1460 CLS
1470 PRINTTAB(15,15);CHR$(141);CHR$(134);"BYE!!!"
1480 PRINTTAB(15);CHR$(141);CHR$(134);"BYE!!!"
1490 FOR X=1 TO 5000:NEXT

```

```
1500 *FX4
1510 *FX11,25
1520 MODE 7
1530 END
```

