

13. 3-D Prisms

General Description

This program has been designed to allow the user to rotate a 3-D prism on the screen. You can create your own prism by altering the data statement in line 2000 (see detailed notes below).

Detailed Description

Lines 5 and 60 First part of delete trick.

10-50 Display title.

70-90 Asks for number of sides to that prism, e.g. a square prism has four sides.

Ask for size to display on screen.

Ask for steps, (this is for rotation).

Note: Validation is not carried out on the above inputs because the user can modify the data line.

100-110 Displays the page giving list of instructions.

120 Converts angle from degrees to radians.

130-180 Sets up variables and reads data for prism.

190 Transfer shape through Z-plane.

200-290 Display each end of prism.

300-360 This is the procedure to transfer shape along Z-plane by two units. (At line 340)

370-410 Join both ends of prism together.

420-570 Get input from keyboard.

580-650 Rotates about Y-axis using rotation matrix.

$$X' = X \cos \Theta - Z \sin \Theta$$

$$Z' = Z \cos \Theta - X \sin \Theta$$

$$Y' = Y$$

660-730 Rotates about X-axis using matrix.

$$X' = X$$

$$Y' = Y \cos \Theta + Z \sin \Theta$$

$$Z' = Z \cos \Theta - X \sin \Theta$$

740-810 Rotate about Z-axis using matrix.

$$X' = X \cos \Theta + Y \sin \Theta$$

$$Y' = Y \cos \Theta - X \sin \Theta$$

$$Z' = Z$$

820-970 Displays help page, and gives a list of instructions.

980-1050 Gives instructions.

1060-1067 Second part of delete trick.

1070-1120 Credits.

2000 Is the data for the prism.

Note: To change shape of prism, change line 2000. The data at line 2000 is in the format

<u> X,Y </u>	,	<u> X,Y </u>	,	<u> X,Y </u>	,	<u> X,Y </u>
First corner		Second corner		Third corner		Fourth corner

Plot shape on graph paper with origin going through middle of the object.

Educational Notes

Whilst this program has an obvious role in clarifying descriptions in geometry lesson, it's also a wonderful demonstration of the potential of computer graphics!

Program Listing

```

5  ONERRORGOTO70
10 TITLE$="3D-PRISMS"
20 MODE 7
30 PRINT TAB(10,5);CHR$(131);CHR$(141);TITLE$
40 PRINT TAB(10,6);CHR$(131);CHR$(141);TITLE$'''
50 A$=INKEY$(400):MODE7
60 PROC_instructions
70 INPUT"HOW MANU SIDES (2-6)",NS
80 INPUT"SIZE (10-300)",S
90 INPUT"IN STEPS OF (1-360)",AN
100 MODE4
110 PROC_HELP
120 AR=RAD(AN)
130 DIM C(2,NS,3)
140 W=600:H=500
150 FOR T=1TONS
160   READ C(1,T,1),C(1,T,2)

```

```

170   C(1,T,3)=-1
180 NEXTT
190 PROC_transfer
200 CLS
210 GCOL0,1
220 FORU=1TO2
230   MOVE W+C(U,1,1)*S,H+C(U,1,2)*S
240   FORK=2TONS
250     DRAWW+C(U,K,1)*S,H+C(U,K,2)*S
260   NEXT
270   DRAWW+C(U,1,1)*S,H+C(U,1,2)*S
280 NEXTU
290 GOTO 370
300 DEF PROC_transfer
310 FORT=1TONS
320   C(2,T,1)=C(1,T,1)
330   C(2,T,2)=C(1,T,2)
340   C(2,T,3)=C(1,T,3)+2
350 NEXT
360 ENDPROC
370 REM DRAW BOTH TOGETHER
380 FOR T=1TONS
390   MOVE W+C(1,T,1)*S,H+C(1,T,2)*S
400   DRAWW+C(2,T,1)*S,H+C(2,T,2)*S
410 NEXTT
420 SOUND1,-10,200,2:AS=GET$
430 *FX15
440 SOUND1,-10,123,2
450 IFA$="" THEN 420
460 IFA$="X" THEN PROC_xaxis
470 IFA$="Y" THEN PROC_yaxis
480 IFA$="Z" THEN PROC_zaxis
490 IFA$="+" THEN S=S+10
500 IFA$="-" THEN S=S-10
510 IFA$="4" THEN MODE4
520 IFA$="5" THEN MODE5
530 IFA$="R" THEN VDU19,1,1,0,0,0
540 IFA$="G" THEN VDU19,1,2,0,0,0
550 IFA$="B" THEN VDU19,1,4,0,0,0
560 IFA$="H" THEN MODE4:PROC_HELP
570 GOTO200
580 DEF PROC_yaxis
590 FORU=1TO2
600   FORT=1TONS
610     X=C(U,T,1):Z=C(U,T,3)
620     C(U,T,1)=(X*COS(AR))-(Z*SIN(AR))
630     C(U,T,3)=(Z*COS(AR))+(X*SIN(AR))
640   NEXTT,U
650 ENDPROC
660 DEF PROC_xaxis
670 FORU=1TO2
680   FORT=1TONS
690     Y=C(U,T,2):Z=C(U,T,3)
700     C(U,T,2)=(Y*COS(AR))+(Z*SIN(AR))
710     C(U,T,3)=(Z*COS(AR))-(Y*SIN(AR))
720   NEXTT,U
730 ENDPROC
740 DEF PROC_zaxis
750 FORU=1TO2
760   FORT=1TONS
770     Y=C(U,T,2):X=C(U,T,1)
780     C(U,T,2)=(Y*COS(AR))-(X*SIN(AR))
790     C(U,T,1)=(X*COS(AR))+(Y*SIN(AR))
800   NEXTT,U
810 ENDPROC
820 DEF PROC_HELP
830 CLS
840 PRINTTAB(15,3);"3D-PRISMS"
850 PRINT'"X..ROTATE ON X-AXIS"

```

```

860      PRINT"Y..ROTATE ON Y-AXIS"
870      PRINT"Z..ROTATE ON Z-AXIS"
880      PRINT'"+"..INCREMENT SCALE"
890      PRINT'"-"..DECREMENT SCALE"
900      PRINT'"4..CHANGE TO MODE 4"
910      PRINT"5..CHANGE TO MODE 5"
920      PRINT'"R..CHANGE COLOUR TO RED"
930      PRINT"G..CHANGE COLOUR TO GREEN"
940      PRINT"B..CHANGE COLOUR TO BLUE"
950      PRINT'"H..DISPLAY THIS PAGE."
960      A$=GET$
970      ENDPROC
980      DEFPROC_instructions
990      PRINTTAB(10,3);"INSTRUCTIONS"
1000     PRINT'"THIS PROGRAM ALLOWS YOU TO ROTATE"
1010     PRINT'"A 3-D PRISM."
1020     PRINT'"IF YOU WISH, YOU CAN MAKE YOUR OWN PRISM"
1030     PRINT'"BY CHANGING THE DATA AT LINE 2000"
1040     PRINT'"THE FORMAT IS  x,y,x,y,etc.."
1050     PRINT'"IT IS EXPLAINED MORE FULLY IN THE BOOK."
1060     PRINT'"PRESS FUNCTION KEY 'f0'"
1065     *KEY0DELETE980,1999|MRUN|M
1066     IFINKEY(-33)=0THEN1066
1067     END
1070     *****
1080     *              3D-PRISMS              *
1090     *              WRITTEN BY              *
1100     *      Andrew pusey. FEB 1983      *
1110     *****
1120     DO NOT RENUMBER THIS PROGRAM
2000     DATA -1,-1,-1,1,1,1,1,-1,0,-.5,-1.5,-1.5

```